Assignment 2, Question 3 The Primitive Euclidean Algorithm

Part (a)

```
> a := expand( (x^4-3*x^3*y-x^2-y)*(8*x-4*y+12)*(2*y^2-2) );
```

$$a := 16x^5y^2 - 56x^4y^3 + 24x^3y^4 + 24x^4y^2 - 72x^3y^3 - 16x^5 + 56x^4y - 40x^3y^2 + 8x^2y^3 - 24x^4 + 72x^3y - 24x^2y^2 - 16xy^3 + 8y^4 + 16x^3 - 8x^2y - 24y^3 + 24x^2 + 16xy - 8y^2 + 24y$$

First we calculate the content. We write the polynomial in Z[y][x]

```
> a := collect(a,x);
```

$$a := (16y^2 - 16)x^5 + (-56y^3 + 24y^2 + 56y - 24)x^4 + (24y^4 - 72y^3 - 40y^2 + 72y + 16)x^3 + (8y^3 - 24y^2 - 8y + 24)x^2 + (-16y^3 + 16y)x + 8y^4 - 24y^3 - 8y^2 + 24y$$

```
> cax := coeff(a,x,5);
```

$$cax := 16y^2 - 16$$

```
> for i from 4 by -1 to 0 do
      c := coeff(a,x,i);
      cax := gcd(cax,coeff(a,x,i));
  od;
```

$$c := -56y^3 + 24y^2 + 56y - 24$$
$$cax := 8y^2 - 8$$
$$c := 24y^4 - 72y^3 - 40y^2 + 72y + 16$$
$$cax := 8y^2 - 8$$
$$c := 8y^3 - 24y^2 - 8y + 24$$
$$cax := 8y^2 - 8$$
$$c := -16y^3 + 16y$$
$$cax := 8y^2 - 8$$
$$c := 8y^4 - 24y^3 - 8y^2 + 24y$$
$$cax := 8y^2 - 8$$

Divide out by the content to get the primitive part

```
> divide( a, cax, 'pax' );
```

$$true$$

```
> collect( pax, x );
```

$$2x^5+(-7y+3)x^4+(3y^2-9y-2)x^3+(y-3)x^2-2yx+y^2-3y$$

Now do this in y

```
> collect(a,y);
```

$$(24x^3+8)y^4+(-56x^4-72x^3+8x^2-16x-24)y^3+(16x^5+24x^4-40x^3-24x^2-8)y^2+(56x^4+72x^3-8x^2+16x+24)y-16x^5-24x^4+16x^3+24x^2$$

```
> cay := coeff(a,y,4):
> for i from 3 by -1 to 0 do cay := gcd(cay,coeff(a,y,i)) od:
> cay;
```

$$8$$

```
> pay := collect( a/cay, y );
```

$$pay := (3x^3+1)y^4+(-7x^4-9x^3+x^2-2x-3)y^3+(2x^5+3x^4-5x^3-3x^2-1)y^2+(7x^4+9x^3-x^2+2x+3)y-2x^5-3x^4+2x^3+3x^2$$

The polynomial a is unit normal in y. So we are done. The unit(a) in y is 1, the content is 8 and the primitive part is pay.

Part (b)

```
> a := 2*x^3-(y+1)*x^2-x+y;
```

$$a := 2x^3-(y+1)x^2-x+y$$

```
> b := (y+2)*x^2-2*x+y;
```

$$b := (y+2)x^2-2x+y$$

We calculate the pseudo-remainder in pr and the pseudo-quotient in pq. The multiplier to use is

```
> m := lcoeff(b,x)^(degree(a,x)-degree(b,x)+1);
```

$$m := (y+2)^2$$

So we divide m a by b .

```
> collect(m*a,x);
```

$$2(y+2)^2x^3+(y+2)^2(-y-1)x^2-(y+2)^2x+(y+2)^2y$$

```
> b := collect(b,x);
```

$$b := (y+2)x^2-2x+y$$

```
> pr := rem( m*a, b, x, 'pq' );
```

$$pr := (-5y^2 - 14y)x + 2y^3 + 7y^2 + 2y$$

```
> pq;
```

$$(2y+4)x - y^2 - 3y + 2$$

```
> expand( m*a-b*pq-pr );
```

$$0$$

Part (c)

```
> a := expand( (x^4-3*x^3*y-x^2-y)*(2*x-y+3)*(8*y^2-8) );
```

$$a := 16x^5y^2 - 56x^4y^3 + 24x^3y^4 + 24x^4y^2 - 72x^3y^3 - 16x^5 + 56x^4y - 40x^3y^2 + 8x^2y^3 - 24x^4 + 72x^3y - 24x^2y^2 - 16xy^3 + 8y^4 + 16x^3 - 8x^2y - 24y^3 + 24x^2 + 16xy - 8y^2 + 24y$$

```
> b := expand( (x^3*y^2+x^3+x^2+3*x+y)*(2*x-y+3)*(12*y^3-12) );
```

$$b := 24x^4y^5 - 12x^3y^6 + 36x^3y^5 + 24x^4y^3 - 12x^3y^4 - 24x^4y^2 + 72x^3y^3 - 12x^2y^4 - 36x^3y^2 + 108x^2y^3 - 12xy^4 - 12y^5 - 24x^4 + 12x^3y + 108xy^3 + 36y^4 - 60x^3 + 12x^2y - 108x^2 + 12xy + 12y^2 - 108x - 36y$$

The contents of a and b wrt x are .

```
> content(a,x,'r0'), content(b,x,'r1');
```

$$8y^2 - 8,\ 12y^3 - 12$$

And their GCD is.

```
> g := gcd( content(a,x), content(b,x) );
```

$$g := 4y - 4$$

The primitive parts are

```
> divide(a,content(a,x),'r0'); collect(r0,x);
```

$$true$$

$$2x^5 + (-7y+3)x^4 + (3y^2 - 9y - 2)x^3 + (y-3)x^2 - 2yx + y^2 - 3y$$

```
> divide(b,content(b,x),'r1'); collect(r1,x);
```

$$true$$

$$(2y^2+2)x^4 + (-y^3 + 3y^2 - y + 5)x^3 + (-y+9)x^2 + (-y+9)x - y^2 + 3y$$

Now we run the primitive PRS using x as the main variable

```
> r2 := prem(r0,r1,x);
```

$$r2 := -8x^3y^4 + 4x^2y^5 + 24x^3y^3 - 24x^2y^4 - 8xy^5 + 4y^6 - 40x^3y^2 + 120x^2y^3 - 8xy^4 - 24y^5 + 24x^3y - 72x^2y^2 + 80xy^3 + 44y^4 - 24x^3 + 112x^2y - 8xy^2 - 36y^3 - 12x^2$$

$$+84\,x\,y+36\,y^2+36\,x$$

```
> r2 := primpart(r2,x);
```

$$r2 := -2\,x^3\,y^4+x^2\,y^5+6\,x^3\,y^3-6\,x^2\,y^4-2\,x\,y^5+y^6-10\,x^3\,y^2+30\,x^2\,y^3-2\,x\,y^4-6\,y^5$$
$$+6\,x^3\,y-18\,x^2\,y^2+20\,x\,y^3+11\,y^4-6\,x^3+28\,x^2\,y-2\,x\,y^2-9\,y^3-3\,x^2+21\,x\,y+9\,y^2$$
$$+9\,x$$

```
> r3 := prem(r1,r2,x);
```

$$r3 := -8\,x^2\,y^{11}+4\,x\,y^{12}+48\,x^2\,y^{10}-36\,x\,y^{11}-136\,x^2\,y^9+76\,x\,y^{10}+32\,y^{11}+800\,x^2\,y^8$$
$$-412\,x\,y^9-192\,y^{10}-456\,x^2\,y^7+1172\,x\,y^8+416\,y^9+2256\,x^2\,y^6-1236\,x\,y^7-672\,y^8$$
$$-464\,x^2\,y^5+3296\,x\,y^6+1024\,y^7+2664\,x^2\,y^4-1380\,x\,y^5-804\,y^6-64\,x^2\,y^3$$
$$+3900\,x\,y^4+1036\,y^5+1520\,x^2\,y^2-520\,x\,y^3-360\,y^4+72\,x^2\,y+2244\,x\,y^2+504\,y^3$$
$$+360\,x^2-36\,y^2+540\,x+108\,y$$

```
> r3 := primpart(r3,x);
```

$$r3 := -2\,x^2\,y^7+x\,y^8+12\,x^2\,y^6-9\,x\,y^7-30\,x^2\,y^5+17\,x\,y^6+8\,y^7+176\,x^2\,y^4-85\,x\,y^5-48\,y^6$$
$$-52\,x^2\,y^3+258\,x\,y^4+88\,y^5+200\,x^2\,y^2-130\,x\,y^3-72\,y^4+18\,x^2\,y+291\,x\,y^2+72\,y^3$$
$$+90\,x^2-9\,y^2+135\,x+27\,y$$

```
> r4 := prem(r2,r3,x);
```

$$r4 := -8\,x\,y^{19}+4\,y^{20}+120\,x\,y^{18}-72\,y^{19}-832\,x\,y^{17}+596\,y^{18}+3624\,x\,y^{16}-3060\,y^{17}$$
$$-11336\,x\,y^{15}+11104\,y^{16}+27624\,x\,y^{14}-30816\,y^{15}-56896\,x\,y^{13}+69884\,y^{14}$$
$$+103640\,x\,y^{12}-137164\,y^{13}-170432\,x\,y^{11}+240676\,y^{12}+245936\,x\,y^{10}-378616\,y^{11}$$
$$-312352\,x\,y^9+525080\,y^{10}+339752\,x\,y^8-638404\,y^9-326472\,x\,y^7+672864\,y^8$$
$$+259080\,x\,y^6-619248\,y^7-192816\,x\,y^5+485028\,y^6+105048\,x\,y^4-341748\,y^5$$
$$-63720\,x\,y^3+189432\,y^4+18792\,x\,y^2-104976\,y^3-9720\,x\,y+33048\,y^2-14580\,y$$

```
> r4 := primpart(r4,x);
```

$$r4 := -2\,x+y-3$$

```
> r5 := prem(r3,r4,x);
```

$$r5 := 0$$

```
> g := expand(g*r4);
```

$$g := -8\,x\,y+4\,y^2+8\,x-16\,y+12$$

```
> lcoeff(g,x);
```

$$-8\,y+8$$

This is not unit normal. Maple's primpart function does not return a unit normal value,

it only divides out by the content.

```
> g := -g;
```

$$g := 8\,x\,y - 4\,y^2 - 8\,x + 16\,y - 12$$